; GNAT
IF X>127 GOTO LEFT
RIGHT
0 TO SPEEDY TO J
246-X*2 TO SPEEDX
85 TO A
242-X TO RANDOM
RANDOM TO Z
CRAWL
GOSUB EYES
IF J=1 GOTO DOWN
IF Z-X<16 GOTO DOWN
250-X*2 TO SPEEDX
GOTO CRAWL
DOWN
0 TO SPEEDX TO J
246-Y*2 TO SPEEDY
175 TO A
242-Y TO RANDOM
RANDOM TO Z
SCUTTLE
GOSUB EYES
IF J=1 GOTO LEFT
IF Z-Y<16 GOTO LEFT
250-Y*2 TO SPEEDY
GOTO SCUTTLE
LEFT
0 TO SPEEDY TO J
10-X*2 TO SPEEDX
X TO RANDOM
RANDOM+14 TO Z
265 TO A
CREEP
GOSUB EYES
IF J=1 GOTO UP
IF X-Z<16 GOTO UP
6-X*2 TO SPEEDX
GOTO CREEP
UP
0 TO SPEEDX TO J
10-Y*2 TO SPEEDY
355 TO A
Y TO RANDOM
RANDOM+14 TO Z
SCAMPER
GOSUB EYES
IF J=1 GOTO RIGHT
IF Y-Z<16 GOTO RIGHT
6-Y*2 TO SPEEDY
GOTO SCAMPER
EYES
A TO AIM TO RADAR
IF RADAR<0 GOTO BITE
AIM+180 TO AIM TO RADAR
IF RADAR<0 GOTO STING
AIM+190 TO AIM TO RADAR
IF RADAR<0 GOTO BITE
AIM+180 TO AIM TO RADAR
IF RADAR<0 GOTO STING
B+1 TO B
ENDSUB
STING
0 TO SPEEDX TO SPEEDY
20-RADAR TO SHOT
GOTO JUMP
BITE
0 TO SPEEDX TO SPEEDY
0-RADAR TO SHOT
JUMP
0 TO B
1 TO J
ENDSUB